

Casos de Estudio Clase 06

CE06A Realizar la multiplicación de dos números enteros A y B mediante sumas sucesivas, hacer la planilla de cálculo y tres algoritmos con cada estructura repetitiva.


	A	B	C	D
1	S	A	B	i
2	0	13	4	0
3	13			1
4	26			2
5	39			3
6	52			4
7				
8				

=A2+\$B\$2

=D2+1


```

Proceso CE06e1v1
  Definir A,B,S,i Como Entero;
  Leer A,B;
  S <- 0;
  Para i<-1 Hasta B Con Paso 1 Hacer
    S <- S + A;
  FinPara
  Escribir A,'x',B,'=',S;
FinProceso
  
```




Proceso CE06e1v2

```
Definir A,B,S,i Como Entero;  
Leer A,B;  
S<-0;  
i<-1;  
Mientras i<=B Hacer  
    S<-S+A;  
    i<-i+1;  
FinMientras  
Escribir A,'x',B,'=',S;  
FinProceso
```



Proceso CE06e1v3

```
Definir A,B,S,i Como Entero;  
Leer A,B;  
S<-0;  
i<-1;  
Repetir  
    Si i<=B Entonces  
        S<-S+A;  
        i<-i+1;  
    FinSi  
Hasta Que NO (i<=B)  
Escribir A,'x',B,'=',S;  
FinProceso
```




CE06B Diseñar un algoritmo que calcule la serie de Taylor de la función seno(x), donde x es el valor de un ángulo (expresado en radianes) y n es el número de términos. Resolver y hacer la prueba de escritorio para 45° y 4 términos.

$$\sin x = \sum_{n=0}^{\infty} \frac{(-1)^n}{(2n+1)!} x^{2n+1}, \forall x$$

$$\text{sen}(x) = x - \frac{x^3}{3!} + \frac{x^5}{5!} - \frac{x^7}{7!} \dots$$

Proceso CE06Bseno

```
// Calculo de la serie de Taylor para la funcion seno
Escribir 'Ingrese grados sexagesimales'
Leer G // angulo en sexagesimales
X <- G/180*PI // angulo en radianes
Escribir 'Ingrese nro de terminos'
Leer Vf
S <- 0
Para n <- 0 Hasta Vf Con Paso 1 Hacer
    K <- 2*n+1 // auxiliar
    F <- 1 // factorial
    Para i <- 1 Hasta K Con Paso 1 Hacer
        F <- F*i
    Fin Para // fin del factorial
    S <- S + (-1)^n / F * X^K
Fin Para
Escribir 'Serie=', S
Escribir 'F(x)=', SEN(X) //funcion interna seno(x)
FinProceso
```



CE06C Hacer una planilla de cálculo que indique la cantidad de divisores que tiene un número entero A. Luego diseñar un algoritmo que resuelva el siguiente enunciado: Ingresar repetidamente un número entero positivo A, encontrar todos sus divisores, finalmente mostrar cuál es el número que tiene el mayor número de divisores. El ingreso termina cuando se ingresa en A un cero.

Planilla divisores de A				
	A	B	C	D
1	A	i	A-ENTERO(A/i)*i	Cont
2	12	1	0	1
3		2	0	2
4		3	0	3
5		4	0	4
6		5	2	4
7		6	0	5
8		7	5	5
9		8	4	5
10		9	3	5
11		10	2	5
12		11	1	5
13		12	0	6

Proceso CE06C

```

Definir A,Cd,Cmax,Amax,i,R Como Entero;
Definir Bmax Como Logico;
Bmax <- Verdadero;
Repetir
    Escribir 'Ingrese A';
    Leer A;
    Cd <- 0;
    Escribir 'Divisor de ',A;
    Para i<-1 Hasta A Con Paso 1 Hacer
        R <- A mod i;
        Si R=0 Entonces
            Cd <- Cd + 1;
            Escribir Sin Saltar i',';
        FinSi
    FinPara
    Escribir ' ';
    Si Bmax Entonces
        Amax <- A;
        Cmax <- Cd;
        Bmax <- Falso;
    Sino
        Si Cd > Cmax Entonces
            Cmax <- Cd;
            Amax <- A;
        FinSi
    FinSi
Hasta Que A=0;
Escribir 'Mayor cantidad de divisores';
Escribir Amax,' con ', Cmax,' divisores' ;
FinProceso

```

